



The World's Largest Open Access Agricultural & Applied Economics Digital Library

This document is discoverable and free to researchers across the globe due to the work of AgEcon Search.

Help ensure our sustainability.

Give to AgEcon Search

AgEcon Search

<http://ageconsearch.umn.edu>

aesearch@umn.edu

*Papers downloaded from **AgEcon Search** may be used for non-commercial purposes and personal study only. No other use, including posting to another Internet site, is permitted without permission from the copyright owner (not AgEcon Search), or as allowed under the provisions of Fair Use, U.S. Copyright Act, Title 17 U.S.C.*

No endorsement of AgEcon Search or its fundraising activities by the author(s) of the following work or their employer(s) is intended or implied.

The Stata Journal (2017)
17, Number 2, pp. 511–514

Stata tip 127: Use capture noisily groups

Roger B. Newson
Department of Primary Care and Public Health
Imperial College London
London, UK
r.newson@imperial.ac.uk

1 The main idea

Do you know about the `capture noisily` group? It is a group of Stata commands, starting with `capture noisily {` and ending with `}`. The commands in between are executed as usual, producing the standard output (because of `noisily`). If any of these commands fail, then execution resumes with the command immediately following the group (because of `capture`). If you do not like typing `capture noisily`, then you can abbreviate it to `cap noi`, or even to `cap n`.

A simple application is where the user wishes to generate a Stata log file, which the user inspects afterward to see whether the logged commands work (and especially if they do not). In a do-file, the user may open the log file and begin the `capture noisily` block, using the commands

```
log using mylog.log, replace
capture noisily {
```

and then add a sequence of Stata commands, such as

```
sysuse auto, clear
regress mpg weight
predict mpghat
twoway scatter mpg weight || line mpghat weight, sort
```

and then end the `capture noisily` block and close the log file, using the commands

```
}
log close
```

The commands inside the block will then be executed until one of them fails (or until all of them end execution, if none fail), and their output will be stored in the file `mylog.log`. Whether or not any of the intervening commands fail, the log file `mylog.log` will be closed by the `log close` command. The user may then inspect the log file with a text editor and view the results if execution was successful or find out what went wrong otherwise. Typically, the number of Stata commands inside the block will be more than the four used here, and there may be program loops and other complicated programming constructions (see [P] **forvalues** and [P] **foreach**), increasing the probability of a failure somewhere.

2 Application in estimation command files

The `capture noisily` prefix is commonly used in do-files containing sequences of estimation commands. If the user is worried that one or more of them might fail (possibly because of insufficient observations), then the user may add a `capture noisily` prefix to each estimation command so that if one estimation command fails, then Stata will resume execution, starting with the next estimation command. If each estimation command is followed by one or more postestimation commands (such as `predict` or `margins`), then each estimation command, and its own subsequent postestimation commands, may be placed in its own `capture noisily` group. That way, if either the estimation command or the postestimation commands fail, then Stata will continue to the next estimation command.

For instance, in `auto.dta`, a user might want to fit a regression model of mileage (`mpg`) with respect to each of the car-size variables `weight`, `length`, and `displacement`, together with the factor `foreign`, indicating whether a car model is made by a non-U.S. company. After each regression model, the user might want to estimate the mean mileages expected if all cars were U.S. models and if all cars were non-U.S. models, assuming that the car-size variable was distributed as in the real-world sample. The code to do this might be as follows:

```
sysuse auto, clear
describe, full
foreach X of var weight length displacement {
    capture noisily {
        regress mpg ibn.foreign `X', noconst vce(robust)
        margins i.foreign
    }
}
```

As it happens, this code executes without any failed commands (not shown). However, if (for any reason) the analysis with respect to `weight` had failed, either in the `regress` command or in the `margins` command, then Stata would have proceeded to the analysis with respect to `length`. If the analysis with respect to `length` had failed, then Stata would have proceeded to the analysis with respect to `displacement`. This feature of `capture noisily` blocks can be very useful if the user is executing a long list of multistep analyses, especially if these analyses involve commands with a higher failure probability than `regress`. Note that if a multistep analysis fails at an earlier command in a `capture noisily` block, then the later commands in the same `capture noisily` block are not attempted. Note also that if the multiple analyses are simply the same command executed on multiple by-groups, then the user does not need the `capture noisily` block, because the user can use `statsby` (see [D] `statsby`).

3 Application in file-generation programs

`capture noisily` may also be used to good effect in file-generation programs. For example, it is used internally by the `dolog`, `dologx`, and `dotex` packages, which can be downloaded from the Statistical Software Components (SSC) archive, and used to execute a do-file while automatically generating a log file. However, more advanced users may want to write output to an arbitrary file, using the `file` command documented in [P] `file`. For instance, the new file may be a `TeX`, an `HTML`, an `Extensible Markup Language`, or a `Rich Text Format` file, produced as an automatically generated report for a reproducible-research project. The user may be using a sequence of commands to generate this new file and may want to close the file after executing those commands, whether or not they all work. The generated file will then be available for the user to inspect (although it may be incomplete), and the user will not have to close it manually. The `capture` block may begin with the commands

```
tempname buff1
file open `buff1' using "myoutput.txt", write text replace
capture noisily {
```

and contain any amount of intervening code, including `file write` statements, such as

```
file write `buff1' "Hello, world!!!!"
```

and end with the commands

```
}
file close `buff1'
```

In this case, a new file `myoutput.txt` is created with a buffer, whose name is stored in the local macro `buff1`, and filled with output from the intervening `file write` statements. However, if any statement in the intervening code fails, then Stata executes the `file close` statement, and the new file (usually incomplete) is available for the user to inspect.

Alternatively, the `capture noisily` block may be preceded and followed by file opening and closing commands other than `file open` and `file close`. For instance, if the user is generating an `HTML` file, then the file opening and closing commands might be the `htopen` and `htclose` commands of the `ht` package (see Quintó et al. [2012]) or the `htmlopen` and `htmlclose` commands of the SSC package `htmlutil` (see Newson [2015]). Or if the user is generating a `Rich Text Format` file, then they might be the `rtfopen` and `rtfclose` commands of the SSC package `rtfutil` (see Newson [2012]). And more user-written file-generating packages are likely to be written on similar lines in the future, possibly for generating files in `Extensible Markup Language`-based document formats yet to be invented. Such future packages are likely to contain their own file-opening and file-closing commands, suitable for use before and after a `capture noisily` block.

4 References

- Newson, R. 2015. `htmlutil`: Stata module to provide utilities for writing hypertext markup language (HTML) files. Statistical Software Components S458085, Department of Economics, Boston College.
<https://ideas.repec.org/c/boc/bocode/s458085.html>.
- Newson, R. B. 2012. From `resultssets` to `resultstables` in Stata. *Stata Journal* 12: 191–213.
- Quintó, L., S. Sanz, E. De Lazzari, and J. J. Aponte. 2012. HTML output in Stata. *Stata Journal* 12: 702–717.